The Focus

December 21. 2005 A Publication for ANSYS Users Issue 43

Modal Analysis in Bite Sized Chunks: CMS

By: Eric Miller

The scene is typical for many large companies that have standardized on ANSYS. Every year management looks at the software renewal budget and sees that lone seat of NASTRAN. They make a call to see why it needs to be renewed and a crusty "guru" emerges from his cubicle, blinking at the brightness of the lights and shying away from contact with humans. A slight funk follows him as he enters the boss's office and states: "we have to keep that seat. ANSYS can't do CMS" He turns and stumbles, his eyes never leaving the ground, back to the safety of his cozy home, surrounded by yellowing stacks of printout and his most prized possession, the aged MSC manuals from the early 80's.

But extinction for this rare breed is forthcoming. With little fanfare, ANSYS began to add Component Mode Synthesis (CMS) to its growing list of features that were once the domain of the NASTRAN user. CMS is a very clever way to conduct an efficient and accurate simulation of the modal behavior of large and complex structures by breaking the structure into pieces, solving them, then combining the results together at the system level. It is a very common technique for any large frame structure such as automobiles, airplanes, rockets and satellites. CMS is basically the extension of sub-structuring from stress to modal analysis.

If you remember sub-structuring (if you don't you should, it can save hours if not days of simulation time) what you do is define nodes at an interface, and use Guyan reduction to create a reduced stiffness matrix in terms of those interface nodes. These nodes are given the politically incorrect name of Master Degrees of Freedom, or MDOF. In CMS the same reduction is done but it includes the additional step of conducting a modal analysis and using the results to modify the transformation matrix that converts stiffness into the reduced matrix. See the theory manual, 17.6.5, for a good description of the process and the actual equations used. This also includes a bibliography of the basic work done to develop this method.

Once the sub-structures are created, you assemble them together at the system level in what is called the "use" pass. At this point you conduct a modal analysis and obtain the "modal" response of the entire system by "synthesizing" it from the "component" responses.



Another difference between CMS and basic sub-structuring is that you have to determine what type of boundary condition you want to force on the MDOF's. When the generation pass does the modal analysis, you can leave the MDOF's free (called free-free) or you can force zero displacement on them (called fixed-fixed). The free-free is more accurate for predicting middle and high frequency modes and the fixed-fixed for lower modes. There are other methods used in the NASTRAN world that ANSYS will be adding per user requests.

After conducting the "use" pass you can view the system response by plotting system mode shapes and looking at modes. If needed, you can also move on to an "expansion" pass and get the response of each DOF in a component at any mode you want. Once you have expanded modes of interest, you can combine the results in the post processor and view the entire system response, or any subset of components, as a system in POST1.

This process is shown in the included example "The CMS Bug" We don't have any models of rockets, airplanes or cars that we can share with the world, so we made a



simple 6 limbed bug with a head and a tail, made out of beams so it runs fast. As you can see by following the APDL scripts, much of the process is repeatable and can be easily automated with generic macros. As you can see in the table, the results line up very well.

Even though it is still developing, the ANSYS implementation of CMS exceeds that of the NASTRAN world in a couple of ways. CMS analysis is pretty simple if your structure is simple, but can get confusing fast for complex structures. ANSYS provides tools to help with bookkeeping as well as copying and translating components in the assembly coordinate system. This seems trivial but can make analysis much easier. In addition, as with anything in ANSYS, you can script the whole thing with APDL.

So, next time you have a large system you need to do a modal analysis on, consider using CMS. The documentation is very clear and well organized. As usual, start with a simple model and use APDL heavily.

(Continued on Page 2)

Contents



(CMS, cont.)

Also, when that NASTRAN guy emerges from his cubicle, let him know he should be looking for a nice retirement community. In a couple more releases ANSYS will have even stronger and easier to use capabilities in this area.



Comparison of Modal Results: Full Model vs. CMS Model

Full Bug (Hz)	CMS Bug (Hz)	Delta (Hz)	Delta (%)
40.52	40.41	-0.11	-0.27%
65.44	65.11	-0.33	-0.50%
75.94	75.77	-0.17	-0.23%
85.71	85.42	-0.28	-0.33%
86.51	86.35	-0.16	-0.18%
101.5	101.24	-0.25	-0.25%
102.99	102.72	-0.28	-0.27%
111.26	111.04	-0.23	-0.21%
111.4	111.18	-0.22	-0.20%
113.75	113.62	-0.13	-0.11%
120.03	119.85	-0.17	-0.15%
122.57	122.37	-0.21	-0.17%
125.58	125.31	-0.27	-0.21%
129.77	129.52	-0.25	-0.19%
131.71	131.21	-0.5	-0.38%
143.16	142.82	-0.35	-0.24%
150.69	150.45	-0.24	-0.16%
171.04	170.51	-0.53	-0.31%
175.51	175.25	-0.27	-0.15%

News

- ANSYS Workbench Integrated With CoCreate CAD
- ANSYS and VISTAGY Form Strategy for Composites
- ANSYS Releases new AWQA and ASAS Versions

FE Term of the Month: Locking

By: Professor F. E. Jargon

Hello friends. I apologize for my extended absence. Things get pretty busy here at the jargon school on occasion. Just because I'm out for a couple of months doesn't mean I'm gone — oh no. I do appreciate your letters of concern, although I'm wondering how some of you got my phone number. As to the person who showed up on my front patio at three in the morning, obviously inebriated—not cool.

Professor F. E. Jargon

A logical follow up to my last article is the topic of locking, by which I mean shear locking and volumetric locking. Both of these items, while trivial as far as a linear analysis is concerned, become quite critical when one is performing a nonlinear analysis. To neglect the possibility of either form of locking can give you erroneous results. In fact, if locking is present in your model, your results will be *wrong*. If you are performing analyses with material or geometric nonlinearities, you must ensure that you have eliminated both forms of locking.

Shear locking is the first form of locking I'll discuss here. Imagine a four-node rectangular element (e.g. SOLID45). Bend the element inward so that it forms a trapezoidal shape as shown in Figure 1. A "real life" object undergoing this sort of bending is going to experience curvature along its top and bottom surfaces. However, with the four-noded quad, the top and bottom surfaces remain straight, inducing artificial shear stresses along the left and right boundaries. For small deflections, this isn't much of an issue. However, with large deflections, shear locking can result in over-stiffening of the model and therefore under-conservative stresses. To alleviate this problem it is wise to choose elements with midside nodes or changing the element formulation. Using the 18x series is highly recommended. These elements have five different forumulations available, B-Bar(default for low order elements), URI (default for high-order elements), Enhanced Strain, Simplified Enhanced Strain, and Mixed u-P. Learn more about these options in PADT's Advanced Structural Nonlinearities training class.



Figure 1: Shapes that Cause Locking

Volumetric locking can affect a plastic or hyperelastic analysis. ANSYS calculates hydrostatic and deviatoric stresses and combines them to produce the total stress state of the model. The problem occurs when calculating hydrostatic stresses. When you have plasticity or hyperelasticity, you're dealing with a material that is mostly incompressible, i.e. your Poisson's ratio is approaching 0.5. When this happens, the equation used to calculate von Mises strains approaches a zero denominator, resulting in mathematical errors in the stress calculation and, again, over-stiffening of the model. To check for the existence of volumetric locking, plot the hydrostatic stresses and look for a checkerboard pattern as shown in Figure 2. To remedy this, consider using Mixed u-P formulation (where 'u' represents displacements and 'P' represents hydrostatic stresses or pressures), so that hydrostatic pressures are calculated as a degree of freedom, rather than being a derived quantity. You may also consider the other element formulations, with them each having pros and cons.



Figure 2: Hydrostatic Stresses

I hope that you've not only received a vocabulary lesson but also some tips on performing a better nonlinear analysis. For even more nonlinear analysis help, check out the Focus archive for numerous articles related to nonlinear modeling or consider taking a training course.

APDL Macros for BUG CMS Example Problem

```
!=======================
!RUNBUG.MAC
!=======================
! Creates BUG full and CMS models
! No Args
!
finish
!------------ FULL BUG MODEL ---------------
/clear             !start with an empty model
/file,fullbug
mkbug           !Run macro that builds model
save
/solu           !Run a standard modal on the
full model
antype,modal
cpintf,all
modopt,lanb,30,0,20000,,off
mxpand,30,,,1
solve
save
finish
/post1
wrt_frq,'fullbug'  !Write frequencies to a file
so we can compare
/prep7             !Grab CP'd nodes, they are
the interface nodes.
nsel,s,cp,,1,9999
cpdel,all,all
cm,nmst,node       ! make a comp called NMST
finish
do_sub,'body'      ! Build substructure of each
part of the model
do_sub,'head'
do_sub,'tail'
do_sub,'leg'
/clear             ! Now use the substructures
to solve
/filnam,bug_use
/prep7
et,1,matrix50      ! define substructure ET
se,body            ! Include each substructure
se,head
se,tail
se,leg
cpintf,all         ! Couple the interface nodes
finish             ! Do a normal modal analysis
/solu
antype,modal
modopt,lanb,30,0,20000,,off
mxpand,30,,,1
solve
finish
/post1
wrt_frq,'cmsbug'   ! Write the freq's out to a
file
save
finish           !---- Now go to each part and
expand
/clear           !     the mode shapes
do_exp,'body'
finish $/clear
do_exp,'head'
finish $/clear
do_exp,'tail'
finish $/clear
do_exp,'leg'
!------- Time to plot the whole model
/post1
cmsfile,add,body,rst    ! Specify the expanded
mode shapes
cmsfile,add,head,rst    !   for each part in the
model
cmsfile,add,tail,rst
cmsfile,add,leg,rst
plt_mshp                ! Make plots

!=======================
!MKBUG.MAC
! Builds bug model.  Change values below to
change size
! No Args
!=======================
/prep7
! Define locations
y1 = -2 $y2 = 0 $y3 = 2
x1 = -10 $x2 = -8 $x3 = -4 $x4 = 0 $x5 = 4
z1 = -1 $z2 = 0 $z3 = 1 $z4 = 2 $z5 = 4
! Mat props
ex,1,10e6
nuxy,1,.23
dens,1,.001
dd1 = (z3-z1)/2 !Set a size variable, DD1 to 1/2
the body height
esize,dd1/2      !Set element size to be 1/2 DD1
!-------------- Make Body
k,1,x2,y2,z3
k,2,x2,y3,z2
k,3,x2,y2,z1
k,4,x2,y1,z2
kgen,2,1,4,,x3-x2
kgen,2,5,8,,x4-x3
*do,i,1,12,4
  l,i,i+1
  l,i+1,i+2
  l,i+2,i+3
  l,i+3,i
*enddo
*do,i,1,4
  l,i,4+i
  l,4+i,8+i
*enddo
sectype,1,beam,csolid
secdata,dd1/15,5,5
et,1,188
lmesh,all
cm,ebody,elem       ! Save as component ebody
esel,u,,,all
!----------------- Make Head
ksel,s,,,1,9,8
ksel,a,,,2,10,4
ksel,a,,,4,12,4
nslk,s,1
cm,nmst,node
k,101,x4,y1,z2
k,102,x4,y2,z1
k,103,x4,y3,z2
k,104,x5,y2,z4
k,105,x5,(y2+y1)/2,z5
k,106,x5,(y3+y2)/2,z5
k,107,(x4+x5)/2,y2,z2
l,101,104
bsplin,102,107,104
l,103,104
l,104,105
l,104,106
sectype,101,beam,csolid
secdata,dd1/15,5,5
secnum,101
lmesh,all
cm,ehead,elem  !Save as component EHEAD
esel,u,,,all
!------------------Make Tail
ksel,s,,,101,103
nslk,s,1
cmsel,a,nmst
cm,nmst,node
k,201,x2,y2,z3
k,202,x1,y2,.8*z4
k,203,x1,y2,.75*z5
k,204,x2,y2,1.5*z5
bspl,201,202,203,204,,,1,0,0,1,0,0

sectype,201,beam,csolid
secdata,dd1/10,5,5
sectype,202,beam,csolid
secdata,dd1/30,5,5
sectype,203,taper
secdata,201,kx(201),ky(201),kz(201)
secdata,202,kx(204),ky(204),kz(204)
secnum,203
lmesh,all
cm,etail,elem  !Save as component ETAIL
esel,u,,,all
!----------- Make legs
ksel,s,,,1
nslk,s,1
cmsel,a,nmst
cm,nmst,node
k,301,x2,y1,z2
k,302,x2,y1*2,z1
k,303,x2,y1*2,z1*4
k,304,x2,y3,z2
k,305,x2,y3*2,z1
k,306,x2,y3*2,z1*4
lsel,u,,,all
bspl,301,302,303
cm,lg1,line
lsel,u,,,all
bspl,304,305,306
cm,lg2,line
sectype,301,beam,csolid
secdata,dd1/10,5,5
sectype,302,beam,csolid
secdata,dd1/25,5,5
sectype,303,taper
secdata,301,kx(301),ky(301),kz(301)
secdata,302,kx(303),ky(303),kz(303)
sectype,304,taper
secdata,301,kx(304),ky(304),kz(304)
secdata,302,kx(306),ky(306),kz(306)
secnum,303
cmsel,a,lg1
lmesh,lg1
secnum,304
lmesh,lg2
lgen,3,all,,,x3-x2
cm,eleg,elem !Save as component ELEG
esel,all
ksel,s,,,1
nslk,s,1
cmsel,a,nmst
cm,nmst,node
allsel
/vup,1,z
/view,1,1,1,1
/eshape,1
Eplot

!==========================
! DO_SUB.MAC
!==========================
! Generic maro to make a substrcture based on
the name in ARG1
! Master DOF's must be defined in component NMST
! Elements must be defined in component E%ARG1%
/filnam,%arg1%
/solu
antype,substr      ! Going to do a substructure
seopt,%arg1%,2     ! Sepcify the name (arg1) and
that mass matrix should be there
cmsopt,fix,30        !  Use fixed BC's on
interface, 30 modes
cmsel,s,e%arg1%      !  Select the element
componenet
nsle
cmsel,r,nmst
m,all,all
nsle
solve
finish
Save

!===================
! DO_EXP.MAC
!===================
! Expands the mode shapes (7-30) for the
substructured parts
! ARG1 is the substructure root name
/filnam,%arg1%
resume
/solu
expass,on
seexp,%arg1%,bug_use
*do,i,7,30
  expsol,1,i
  solve
*enddo
finish
!=======================
!WRT_FRQ.MAC
! par1 = root name for file
!=======================
! Simple utility macro to output frequencies
*get,mxstp,active,,set,nset
*cfopen,%arg1%.txt
*vwrite,mxstp
%g
*do,i,1,mxstp
  set,next
  *get,frq,active,,set,freq
  *vwrite,frq
%g
*enddo
*cfclose

!===================
! PLT_MSHP.MAC
!===================
! Simple macro to plot out modeshapes
allsel
/plopt,info,2 $/plopt,leg3,off $/plopt,minm,off
/eshape,1,
/win,1,ltop  $/win,2,rtop  $/win,3,lbot
$/win,4,rbot
/view,1,0,0,1 $/vup,1,y
/view,2,1,0,0 $/vup,2,z
/view,3,0,-y,0 $/vup,3,z
/view,4,1,1,1  $vup,4,z
*get,mxstp,active,,set,nset
*do,i,1,mxstp
  set,i
  plnsol,u,sum,2
*enddo
```

PCG Solver Improvement in V10.0

By: Rod Scholl



With all the speed enhancements over the last couple years to the solution phase, solve time has become much less of a consideration when doing FEA – and that is really sayin' something.

Previously I used PCG solver as a rule, and only used the Sparse for the few types of analyses I did that required it – and the occasional analysis where the number of PCG iterations exceeded 1000; on about 1 in 5 analyses I also would switch to sparse after getting this message that perhaps you are familiar with. Occasionally changing the MULT option on EQSLV would help as it suggests… but usually not.

It could be the type of analyses I did in 2004-2005 were mostly ill-conditioned – or maybe the continued speed increases of element formulation and other steps just made me pleased to be solving faster than last year and I abandoned the 1.5X-3X benefit I usually saw using PCG over the Sparse. I do know that there was just something disturbing about watching the iterations climb to 1500+, and then debate whether to bail out now and switch to sparse or continue waiting for the PCG to handle the poorly conditioned problem. I don't like such weighty decisions… I suppose I began to avoid the dilemma by leaving off the EQSLV,PCG command…

However, in 10.0 the PCG solver has been made even more robust. I've done some testing with it, and I am again switching back to it as the preferred solver. (It's really a matter of choosing between two great choices, which is a nice luxury.)

It took me a little foolin' around to make a macro of an ill-conditioned problem. This macro fails during PCG iterations in version 9.0 – but works like a charm in 10.0. If you want to revert to 9.0 behavior (I'm not sure for what reason other than testing) use the PCGOPT,1 command. The default is PCGOPT,0 which lets ANSYS decide the stability vs. solution time dilemma, and in my testing seemed the best way to go. But if you get into a particularly poorly conditioned problem try PCGOPT,4 – I suspect it would still give the sparse solver a run for its money.

Thus my estimate is the above error message will quickly be a thing of the past.

Links

Want to learn more about the programming language used to make the ANSYS GUI? Want to customize the GUI? www.padtinc.com/support/ansys/tcltk

Join the growing Workbench user community on the ANSYS, Inc. customer portal. www1.ansys.com/customer Log in and navigate to the WB Community

Upcoming Training Classes

Month	Start	End	#	Title	Location
Jan '06	17-Jan	19-Jan	104	ANSYS Workbench Simulation - Intro	Irvine, CA
	20-Jan	20-Jan	105	ANSYS Workbench Sim Struct Nonlin	Irvine, CA
	26-Jan	27-Jan	801	ANSYS Customization with APDL	Tempe, AZ
Feb '06	6-Feb	8-Feb	101	Introduction to ANSYS, Part I	Irvine, CA
	9-Feb	10-Feb	202	Advanced Structural Nonlinearities	Tempe, AZ
	13-Feb	15-Feb	104	ANSYS Workbench Simulation - Intro	Tempe, AZ
	16-Feb	16-Feb	105	ANSYS Workbench Sim Struct Nonlin	Tempe, AZ
	22-Feb	23-Feb	301	Heat Transfer	Tempe, AZ
Mar '06	6-Mar	8-Mar	101	Introduction to ANSYS, Part I	Albq, NM
	9-Mar	10-Mar	102	Introduction to ANSYS, Part II	Albq., NM
	16-Mar	17-Mar	203	Dynamics	Tempe, AZ
	23-Mar	24-Mar	501	ANSYS/LS-DYNA	Irvine, CA
	27-Mar	29-Mar	902	Multiphysics Simulation for MEMS	Tempe, AZ

Resources

PADT's "ANSYS Customization with APDL" is a great way to learn all about macro writing for ANSYS

Need Material Properties? MatWeb is a huge online database of material properties that now supports ANSYS file formats.

The Focus is a periodic publication of Phoenix Analysis & Design Technologies (PADT). Its goal is to educate and entertain the worldwide ANSYS user community. More information on this publication can be found at: http://www.padtinc.com/epubs/focus/about